/**
 * 不规则碰撞检测
 * @author http://www.tink.ws/blog/as-30-hittest
 * 2009-11-12 9:54
 */
package com.ibio8.components.helpers {
	import flash.display.BitmapData;
	import flash.display.BlendMode;
	import flash.display.DisplayObject;
	import flash.display.Sprite;
	import flash.geom.ColorTransform;
	import flash.geom.Matrix;
	import flash.geom.Point;
	import flash.geom.Rectangle;
	/**
	 * 不规则碰撞检测。
	 * 
	 * <p>
	 * 在网上找的一个解决方案，用来进行对不规则碰撞物体的检测。最常见的用法就是：
	 * 对 PNG 图片透明区域的忽略，来进行碰撞检测。
	 * </p>
	 * <b>此类来自网络，作者：http://www.tink.ws/blog/as-30-hittest</b>
	 */
	public class HitTest {
		
		/**
		 * 检测输入的两个显示对象
		 * @param	target1		显示对象一
		 * @param	target2		显示对象二
		 * @param	accuracy	精确度，默认为 1
		 * @return				是否碰撞
		 */
		public static function complexHitTestObject(target1:DisplayObject, target2:DisplayObject, accuracy:Number = 1):Boolean {
			return complexIntersectionRectangle(target1, target2, accuracy).width != 0;
		}
		
		/** @private **/
		public static function intersectionRectangle(target1:DisplayObject, target2:DisplayObject):Rectangle {
			//If either of the items don't have a reference to stage, then they are not in a display list
			//or if a simple hitTestObject is false, they cannot be intersecting.
			if ((!target1.root) || (!target2.root) || (!target1.hitTestObject(target2))) {
				return new Rectangle();
			}
			//Get the bounds of each DisplayObject.
			var bounds1:Rectangle = target1.getBounds(target1.root);
			var bounds2:Rectangle = target2.getBounds(target2.root);
			//找到两个接触物体的交集
			var intersection:Rectangle = new Rectangle();
			intersection.x = Math.max(bounds1.x, bounds2.x);
			intersection.y = Math.max(bounds1.y, bounds2.y);
			intersection.width = Math.min((bounds1.x + bounds1.width) - intersection.x, (bounds2.x + bounds2.width) - intersection.x);
			intersection.height = Math.min((bounds1.y + bounds1.height) - intersection.y, (bounds2.y + bounds2.height) - intersection.y);
			return intersection;
		}
		
		/** @private **/
		public static function complexIntersectionRectangle(target1:DisplayObject, target2:DisplayObject, accuracy:Number = 1):Rectangle {
			if (accuracy <= 0) {
				throw new Error("ArgumentError: Error #5001: Invalid value for accuracy", 5001);
			}
			//If a simple hitTestObject is false, they cannot be intersecting.
			if (!target1.hitTestObject(target2)) {
				return new Rectangle();
			}
			var hitRectangle:Rectangle = intersectionRectangle(target1, target2);
			//If their boundaries are no interesecting, they cannot be intersecting.
			if ((hitRectangle.width * accuracy < 1) || (hitRectangle.height * accuracy < 1)) {
				return new Rectangle();
			}
			var bitmapData:BitmapData = new BitmapData(hitRectangle.width * accuracy, hitRectangle.height * accuracy, false, 0x000000);
			//Draw the first target.
			bitmapData.draw(target1, getDrawMatrix(target1, hitRectangle, accuracy), new ColorTransform(1, 1, 1, 1, 255, -255, -255, 255));
			//Overlay the second target.
			bitmapData.draw(target2, getDrawMatrix(target2, hitRectangle, accuracy), new ColorTransform(1, 1, 1, 1, 255, 255, 255, 255), BlendMode.DIFFERENCE);
			//Find the intersection.
			var intersection:Rectangle = bitmapData.getColorBoundsRect(0xFFFFFFFF, 0xFF00FFFF);
			bitmapData.dispose();
			//Alter width and positions to compensate for accuracy
			if (accuracy != 1) {
				intersection.x /= accuracy;
				intersection.y /= accuracy;
				intersection.width /= accuracy;
				intersection.height /= accuracy;
			}
			intersection.x += hitRectangle.x;
			intersection.y += hitRectangle.y;
			return intersection;
		}
		
		/** @private **/
		protected static function getDrawMatrix(target:DisplayObject, hitRectangle:Rectangle, accuracy:Number):Matrix {
			var localToGlobal:Point;
			var matrix:Matrix;
			var rootConcatenatedMatrix:Matrix = target.root.transform.concatenatedMatrix;
			localToGlobal = target.localToGlobal(new Point());
			matrix = target.transform.concatenatedMatrix;
			matrix.tx = localToGlobal.x - hitRectangle.x;
			matrix.ty = localToGlobal.y - hitRectangle.y;
			matrix.a = matrix.a / rootConcatenatedMatrix.a;
			matrix.d = matrix.d / rootConcatenatedMatrix.d;
			if (accuracy != 1) {
				matrix.scale(accuracy, accuracy);
			}
			return matrix;
		}
	}
}